 
CORRELATION AND REGRESSION
LECTURE 5 Mar 3nd , 2009
 
A. Introduction
 
A common statistic for indicating the strength of a linear relationship existing between two continuous variables is called the Pearson correlation coefficient, or correlation coefficient (there are other types of correlation coefficients). The correlation coefficient is a number ranging from -1 to +1. A positive correlation means that as values of one variable increase, values of the other variable also tend to increase. A small or zero correlation coefficient tells us that the two variables are unrelated. Finally, a negative correlation coefficient show an inverse relationship between the variable: as one goes up, the other goes down.
 
B. Correlation
 
We have recorded the GENDER, HEIGHT, WEIGHT, and AGE of seven people and want to compute the correlations between HEIGHT and WEIGHT, HEIGHT and AGE, and WEIGHT and AGE.
 

DATA CORR_EG;
      INPUT GENDER $ HEIGHT WEIGHT AGE;
DATALINES;
M  68    155   23
F  61    99    20
F  63    115   21
M  70    205   45
M  69    170   .
F  65    125   30
M  72    220   48
;
Run;
PROC CORR DATA=CORR_EG;
      TITLE ‘EXAMPLE OF A CORRELATION MATRIX’;
      VAR HEIGHT WEIGHT AGE;
RUN;
 
 
The output:
 
Pearson Correlation Coefficients
Prob > |r| under H0: Rho=0
Number of Observations
 
                                      HEIGHT        WEIGHT           AGE
 
                        HEIGHT       1.00000       0.97165       0.86614
                                                    0.0003        0.0257
                                           7             7             6
 
                        WEIGHT       0.97165       1.00000       0.92496
                                      0.0003                      0.0082
                                           7             7             6
 
                        AGE          0.86614       0.92496       1.00000
                                      0.0257        0.0082
                                           6             6             6
 
 
C.     Significance of a Correlation Coefficient
 
“How large a correlation coefficient do I need to show that two variables are correlated?” Each time PROC CORR prints a correlation coefficient, it also prints a probability associated with the coefficient. That number gives the probability of obtaining a sample correlation coefficient as large as or larger than the one obtained by chance alone.
 
The significance of a correlation coefficient is a function of the magnitude of the correlation and the sample size. With a large number of data points, even a small correlation coefficient can be significant.
 
It is important to remember that correlation indicates only the strength of a relationship – it does not imply causality. For example, we would probably find a high positive correlation between the number of hospital in each of the 50 states versus the number of household pets in each state. It does not mean that pets make people sick and therefore make more hospitals necessary! The most plausible explanation is that both variables (number of pets and number of hospitals) are related to population size.
 
Being SIGNIFICANT is not the same as being IMPORTANT or STRONG. That is, knowing the significance of the correlation coefficient does not tell us very much. Once we know that our correlation coefficient is significantly different from zero, we need to look further in interpreting the importance of that correlation.
 
 
 
D.  How to Interpret a Correlation Coefficient
 
One of the best ways to interpret a correlation coefficient(r) is to look at the square of the coefficient(r-squared);r-squared can be interpreted as the proportion of variance in one of the variables that can be explained by variation in the other variable. As an example, our height/weight correlation is .97. Thus, r-squared is .94.We can say that 94% of the variation in weights can be explained by variation in height (or vice versa). Also, (1-0.94) or 6% of the variance of weight, is due to factors other than height variation.
 
Another consideration in the interpretation of correlation coefficient is this: Be sure to look at a scatter plot of the data (using PROC PLOT). It often turns out that one or two extreme data points can cause the correlation coefficient to be much larger than expected. One keypunching error can cause dramatically alter a correlation coefficient.
 
An important assumption concerning a correlation coefficient is that each pair of x.y data points is independent of any other pair. That is, each pair of points has to come from a separate subject. Otherwise we can not compute a valid correlation coefficient.
 
 
 
E.   Partial Correlations
 
A researcher may wish to determine the strength of the relationship between two variables when the effect of other variables has been removed. One way to accomplish this is by computing a partial correlation. To remove the effect of one or more variables from a correlation, use a PARTIAL statement to list those variables whose effects you want to remove. 
 

PROC CORR DATA=CORR_EG NOSIMPLE;
   TITLE ‘EXAMPLE OF A PARTIAL CORRELATION’;
   VAR HEIGHT WEIGHT;
   PARTIAL AGE;
RUN;
 
The output:
The CORR Procedure
 
                            1 Partial Variables:    AGE
                            2         Variables:    HEIGHT   WEIGHT
 
 
                        Pearson Partial Correlation Coefficients, N = 6
                               Prob > |r| under H0: Partial Rho=0
 
                                             HEIGHT        WEIGHT
 
                               HEIGHT       1.00000       0.91934
                                                           0.0272
 
                               WEIGHT       0.91934       1.00000
                                             0.0272
 
 
 
F.  Linear Regression
 
Giving a person’s height, what would be their predicted weight? How can we best define the relationship between height and weight? By studying the graph below we see that the relationship is approximately linear. That is, we can imagine drawing a straight line on the graph with most of the data points being only a short distance from the line. The vertical distance from each data point to this line is called a residual.
 
How do we determine the “best” straight line to fit out height/weight data? The method of least squares is commonly used, which finds the line (called the regression line) that minimizes the sum of the squared residuals. An equivalent way to define a residual is the difference between a subject’s predicted score and his/her actual score.
 

PROC REG DATA=CORR_EG;
      TITLE ‘REGRESSON LINE FOR HEIGHT-WEIGHT DATA’;
      MODEL WEIGHT=HEIGHT;
RUN;
 
The output:
The REG Procedure
                                         Model: MODEL1
                                  Dependent Variable: WEIGHT
 
                                      Analysis of Variance
 
                                             Sum of           Mean
         Source                   DF        Squares         Square    F Value    Pr > F
 
         Model                     1          11880          11880      84.45    0.0003
         Error                     5      703.38705      140.67741
         Corrected Total           6          12584
 
 
                      Root MSE             11.86075    R-Square     0.9441
                      Dependent Mean      155.57143    Adj R-Sq     0.9329
                      Coeff Var             7.62399
 
 
                                      Parameter Estimates
 
                                   Parameter       Standard
              Variable     DF       Estimate          Error    t Value    Pr > |t|
 
              Intercept     1     -592.64458       81.54217      -7.27      0.0008
              HEIGHT        1       11.19127        1.21780       9.19      0.0003
 
 
 
G.  Partitioning the Totoal Sum of Squares
 
The total sum of squares is the sum of squared deviations of each person’s weight from the grand mean. This total sum of squares (SS) can be divided into the two portions: the sum of squares due to regression (or model), and the sum of square error (sometimes called residual). One portion called the Sum of Squares (ERROR) in the output, reflects the deviations of each weight from the PREDICTED weight. The other portion reflects deviations between the PREDICTED values and the MEAN.
 
Our intuition tells us that if the deviations about the regression line are small (error mean square) compared to the deviation between the predicted values and the mean (mean square model) then we have a good regression line. To compare these mean squares, the ratio F=Mean square mode/Mean square error is formed. The larger this ratio, the better the fit.
 
 
 
H.  Plotting the Points on the Regression Line
 
In particular, the names PREDICTED. And RESIDUAL. (The periods are part of these keywords) are used to plot predicted and residual values. A common option is OVERLAY, which is used to plot more than one graph on a single set of axes.
 

PROC REG DATA= CORR_EG;
      MODEL WEIGHT=HEIGHT;
      PLOT PREDICTED.*HEIGHT=’P’    WEIGHT*HEIGHT=’*’ / OVERLAY;
RUN;
 

THE OUTPUT:

I.  Plotting Residuals and Confidence Limits
 
The most useful statistics besides the predicted values are:
RESIDUAL. – The residual (i.e., difference between actual and predicted values for each observation).
L95. U95. – The lower or upper 95% confidence limit for an individual predicted value. (i.e., 95% of the dependent data values would be expected to be between these two limits).
L95M. U95M. – The lower or upper 95% confidence limit for the mean of the dependent variable for a given value of the independent variable.
 

PROC REG DATA=CORR_EG;
      MODEL WEIGHT = HEIGHT;
      PLOT  PREDICTED.*HEIGHT = ‘P’
            U95M.*HEIGHT = ‘-‘ L95M.*HEIGHT=’-‘
            WEIGHT*HEIGHT=’*’ / OVERLAY;
      PLOT  RESIDUAL.*HEIGHT = ‘O’;
RUN;
 

the output:

J.  Adding a Quadratic Term to the Regression Equation
 
The plot of residuals, shown in the graph above, suggests that a second-order term (height squared) might improve the model since the points do not seem random but, rather, form a curve that could be fit by a second-order equation. To add a quadratic term, first, we need to add a line in the original DATA step to compute a variable that represents the height squared. After the INPUT statement, include a line such as the following:
 
      HEIGHT2 = HEIGHT * HEIGHT; or HEIGHT2 = HEIGHT**2;
 
Next,write your MODEL and PLOT statements:
 

     PROC REG DATA=CORR_EG;
         MODEL WEIGHT = HEIGHT HEIGHT2;
         PLOT RESIDUAL.*HEIGHT = ‘O’;
   RUN;
          

When you run this model, you will get an r-squared of .9743, and improvement over the .9441 obtained with the linear model.
 

 
 
K.  Transforming Data
 
Another regression example is provided here to demonstrate some additional step that may be necessary when doing regression.
 
DATA HEART;
   INPUT DRUG_DOSE HEART_RATE;
DATALINES;
2  60
2  58
4  63
4  62
8  67
8  65
16 70
16 70
32 74
32 73
;
 
PROC PLOT DATA=HEART;
   PLOT HEART_RATE*DRUG_DOSE=’O’;
RUN;
PROC REG DATA=HEART;
   MODEL HEART_RATE=DRUG_DOSE;
RUN;
 

 
 

Either by clinical judgment or by careful inspection of the graph, we decide that the relationship is not linear. We see an approximately equal increase in heart rate each time the dose is doubled. Therefore, if we plot log dose against heart rate we can expect a linear relationship. SAS software has a number of built-in functions such as logarithms and trigonometric functions. We can write mathematical equations to define new variable by placing these statements between the INPUT and DATALINES statements.
 

DATA HEART;
INPUT DRUG_DOSE HEART_RATE;
   L_DRUG_DOSE = LOG(DRUG_DOSE);
DATALINES;
 
PROC PLOT DATA=HEART;
   PLOT HEART_RATE*L_DRUG_DOSE=’O’;
RUN;
 
PROC REG DATA=HEART;
   MODEL HEART_RATE=L_DRUG_DOSE;
RUN;
 

 
 
Notice that the data points are now closer to the regression line. The MEAN SQUARE ERROR term is smaller and r-square is large, confirming our conclusion that dose versus heart rate fits a logarithmic curve better than a linear one.
 
 
  

